
This Talk

▸ We have been interested in applying formal verification to
information security

▸ We have been led to look at topics such as information theory,
error-correcting codes, probabilistic programs, etc.

▸ We would like to report on a few formal theories that might
be of general interest

▸ The common topic is probability, the work spans several years

▸ Our message is that MathComp has been providing us with
a reliable environment for our experiments

SSReflect Early Adopters?

▸ Starting in 2004, we were working on formal proof of imperative
programs in Coq using separation logic

▸ We were facing productivity issues that we failed to analyze correctly

▸ By chance, we ran into

and realized that many of our concerns were addressed by
SSReflect



SSReflect/MathComp in Japan

1. We have been trying to promote MathComp in Japan with
lectures

2. These lectures turned into a book in 2018

3. In this book, tactics are given mascots, e.g.:



The SSReflect Zoo According to

move view case

rewrite elim apply

Fundamental Questions Answered Information Theory

1. What is the ultimate data compression rate1?

2. What is the ultimate encoding rate2 for communication?

Setting:

▸ a source is a probability distribution

▸ a channel is a stochastic matrix
(e.g., the binary symmetric channel [ 1−p p

p 1−p ])

Shannon answered the fundamental questions in 1948:

1. Source coding theorem: One cannot minimize the
compression rate below the entropy

2. Channel coding theorem: One cannot maximize the encoding
rate beyond the capacity of a channel

1compressed bitstring size / original message size
2original message size / encoded message size



Information Theory: Basic Definitions [CT01]

▸ The entropy of a random variable taking n values with
probabilities pi is H = −∑

n−1
i=0 pi log pi

▸ Given two probability mass functions p and q, the divergence

is D(p∣∣q) = ∑x∈X p(x) log p(x)
q(x)

▸ The mutual information between two random variables X , Y
with joint probability mass function p(x , y) and marginals
p(x) and p(y) is I (X ;Y ) = D(p(x , y)∣∣p(x)p(y))

▸ Consider an input X and an output Y . A channel is a
conditional probability distribution pY ∣X (y ∣x). The capacity
is C = maxp(x) I (X ;Y ), p(x) ranging over all the input
distributions

⇒ It looks like a good fit for SSReflect’s iterated
operators [BGBP08]



Finite Probabilities with SSReflect and Coq
(We are in 2009)

Finite probability theory with the iterated operators and the finite sets of
MathComp and the real numbers of Coq:

▸ Distributions over a finite type:

Record fdist (A : finType) := mk {

f :> A ->R+ ;

_ : \sum_(a in A) f a == 1 :> R }.

▸ Probability of an event E given a distribution P:

Definition Pr P (E : {set A}) := \sum_(a in E) P a.

▸ Random variables:

Definition RV U T (P : fdist U) := U -> T.

▸ Distribution of a random variable X: `Pr[ X = a ],
shortcut for Pr P (X @^-1: [set a]) or for
fdistmap X P a



Information Theory with MathComp
Overview

Starting from:

▸ H = −∑n−1
i=0 pi log pi

Definition entropy := - \sum_(a in A) P a * log (P a).

▸ ∑x∈X p(x) log p(x)
q(x)

Definition div := \sum_(a in A) P a * log (P a / Q a).

▸ D(p(x , y)∣∣p(x)p(y))

Definition mutual_info := D(PQ || PQ`1 `x PQ`2).

We completely formalized an introductory textbook to information
theory:
▸ Shannon’s theorems [AH12, AHS14]

▸ Error-correcting codes (Hamming, BCH,
Reed-Solomon, LDPC) [AG15, AGS20b]

▸ Presentations to information theorists
[OHA14, AGS16, AGS18]



Convexity of Information-theoretic Definitions
Statements from [CT01]

Theorem
H(p) is a concave function of p.

Entropy of a binary distribution

Theorem
D(p∣∣q) is convex in the pair (p,q), i.e., if (p1,q1) and (p2,q2) are pairs
of probability mass functions, then

D(λp1 + (1 − λ)p2∣∣λq1 + (1 − λ)q2) ≤ λD(p1∣∣q1) + (1 − λ)D(p2∣∣q2)

for all 0 ≤ λ ≤ 1.

Theorem
Let (X ,Y ) ∼ p(x , y) = p(x)p(y ∣x). The mutual information I (X ;Y ) is a
concave function of p(x) for fixed p(y ∣x) and a convex function of
p(y ∣x) for fixed p(x).



Convex Space

A convex space is a carrier together with a family of binary operators
a◁ p▷ b with 0 ≤ p ≤ 1 such that [Sto49, Fri09]:

▸ a◁ 0▷ b = b

▸ a◁ p▷ a = a (idempotence)

▸ a◁ p▷ b = b◁ 1 − p▷ a (skewed commutativity)

▸ a◁ p▷ (b◁ q▷ b) = (a◁ r ▷ b)◁ s ▷ c (quasi-associativity)
with s = p̄q̄ and r = p

s
(where x̄ = 1 − x)

Examples: real numbers (pa + (1 − p)b), functions to a convex space,
finite distributions, etc.

Allows for generic definitions (U convex space, V ordered convex space):

▸ f ∶ U → U ′ is affine
def
= ∀a,b,0 ≤ p ≤ 1, f (a◁ p▷ b)=f (a)◁ p▷ f (b)

▸ f ∶ U→V is convex
def
= ∀a,b,0 ≤ p ≤ 1, f (a◁p▷b)≤f (a)◁p▷ f (b)

▸ also convex sets and hulls



Convex Space in MathComp

Conveniently defined using Hierarchy-Builder [CST20]:

1. Declare an interface:
HB.mixin Record isConvexSpace (T : Type) := {

_ <| _ |> _ : forall p, T -> T -> T ;

conv1 : forall a b, a <| 1%:pr |> b = a ;

convmm : forall p a, a <| p |> a = a ;

convC : forall p a b, a <| p |> b = b <| p.~%:pr |> a;

convA : forall (p q : prob) (a b c : T),

a <| p |> (b <| q |> c) =

(a <| [r_of p, q] |> b) <| [s_of p, q] |> c }.

2. Declare a structure:
#[short(type=convType)]

HB.structure Definition ConvexSpace := {T of isConvexSpace T }.

3. Build instances: any lmodType (and thus real numbers), the
type “fdist A”, the type “A -> fdist B”, etc.



Application of Convex Spaces to Information Theory

▸ Short statements for convexity properties of information
theoretic definitions:
▸ Lemma entropy_concave :

concave_function (fun P : fdist A => `H P).

▸ Lemma mutual_information_concave W :

concave_function (fun P => mutual_info (P `X W)).

where P `X W is the product distribution λ(x , y).P x ⋅W x y

▸ Lemma mutual_information_convex P :

convex_function

(fun W : A -> fdist B => mutual_info (P `X W)).



Real Cones
A practical tool to reason about convexity

▸ In a convex space, quasi-associativity and skewed commutativity
make for cumbersome symbolic computations

▸ It is actually possible to transpose such computations into real
cones where addition is commutative and associative [VW06]:

HB.mixin Record isQuasiRealCone A := {

addpt : A -> A -> A ;

zero : A ;

addptC : commutative addpt ;

addptA : associative addpt ;

addpt0 : right_id zero addpt ;

scalept : R -> A -> A ;

scale0pt : forall x, scalept 0 x = zero ;

scale1pt : forall x, scalept 1 x = x ;

scaleptDr : forall r, {morph scalept r : x y / addpt x y >-> addpt x y} ;

scaleptA : forall p q x, 0 <= p -> 0 <= q ->

scalept p (scalept q x) = scalept (p * q) x }.

HB.mixin Record isRealCone A of isQuasiRealCone A := {

scaleptDl : forall p q x, 0 <= p -> 0 <= q ->

scalept (p + q) x = addpt (scalept p x) (scalept q x) }.



From Convex Spaces to Real Cones

Consider the following inductive type:

Inductive scaled (A : Type) := Scaled of Rpos & A | Zero.

When A is a convex space:

▸ scaled A can be equipped with a real cone structure (take
addpt (Scaled r x) (Scaled q y) to be (r + q)(x ◁ r

r+q ▷ y))

▸ scaled A can be equipped with a convex space structure (take
x ◁ p▷ y to be addpt (scalept p x) (scalept (1 - p) y)

[VW06])

We can transpose symbolic computations using the fact that Scaled 1

is injective and affine:

▸ a <|p|> b

→ Scaled 1 a <|p|> Scaled 1 b

→ addpt (scalept p (Scaled 1 a)) (scalept (1 - p) (Scaled 1 b))

where addition is associative and commutative

See InfoTheo online or [AGS20a] for details

Monadic Equational Reasoning

This is an approach to verify programs with effects using
equational reasoning [GH11]
▸ effects are represented by monad interfaces with typically:

▸ an operator (failure, arbitrary choice, probabilistic choice, etc.)
▸ rewriting laws in the form of equations

▸ monad interfaces can inherit from other interfaces and can be
combined

Our starting idea:

▸ build a hierarchy of interfaces using packed classes [GGMR09]

▸ use SSReflect’s rewrite [GT12] to perform equational
reasoning



Example of Monadic Laws
Reminder

Monad laws (two operators: ret(⋅) and ⋅≫= ⋅):

1. ret(a)≫= f = f a (left neutral)

2. m≫= (λx .ret(x)) = m (right neutral)

3. (m≫= f )≫= g = m≫= (λx .f x ≫= g) (associativity)

Arbitrary choice (one operator: ⋅2 ⋅):

1. (m12m2)2m3 = m12 (m22m3) (associativity)

2. (m12m2)≫= k = (m1 ≫= k)2 (m2 ≫= k)
(left-distributivity of bind w.r.t. arbitrary choice)

etc.



Functors and Monads with Hierarchy-Builder

▸ We consider Coq’s Type to be the category Set of sets and
functions [TJ16]

▸ Let us start with functors:
▸ action on objects: F : Type -> Type (carrier)
▸ action on morphisms: actm below

HB.mixin Record isFunctor (F : Type -> Type) := {

actm : forall A B, (A -> B) -> F A -> F B;

functor_id : FunctorLaws.id actm ;

functor_o : FunctorLaws.comp actm }.

▸ Next, monads (ret/bind interface):

HB.factory Record isMonad_ret_bind (F : Type -> Type) := {

ret' : forall A, A -> F A ;

bind : forall A B, F A -> (A -> F B) -> F B ;

bindretf : BindLaws.left_neutral bind ret' ;

bindmret : BindLaws.right_neutral bind ret' ;

bindA : BindLaws.associative bind }.



The Interface of the Probability Monad

Probability monad:

▸ extends the type of Monad

▸ similar interface to convex spaces

▸ with left-distributivity of bind w.r.t. probabilistic choice

HB.mixin Record isMonadProb (M : Type -> Type) of Monad M := {

_ <| _ |> _ : forall p T, M T -> M T -> M T ;

choice0 : forall T a b, a <| 0 |> b = b ;

choiceC : forall T p a b, a <| p |> b = b <| 1 - p |> a ;

choicemm : forall T p, idempotent (_ <| p |> _) ;

choiceA : forall T p q r s a b c,

p = r * s -> 1 - s = (1 - p) * (1 - q) ->

a <| p |> (b <| q |> c) = (a <| r |> b) <| s |> c;

choice_bindDl : forall p a b,

(a <| p |> b) >>= f = (a >>= f) <| p |> (b >>= f) }.



Model of the Probability Monad
The interface do have an implementation

▸ Finite distributions do not form a monad because
fdist : finType -> Type is not an endofunction

▸ Hence finitely-supported distributions with finmap [CS15]:
Record fsdist (A : choiceType) := mk {

f :> {fsfun A -> R with 0} ;

_ : all (fun x => 0 <b f x) (finsupp f) &&

\sum_(a <- finsupp f) f a == 1}.

▸ The required operators (ret(⋅), ⋅≫= ⋅, ⋅ ◁ ⋅ ▷ ⋅):
▸ fsdist1 : forall A : choiceType, A -> {dist A}

def
= [fsfun b in [fset a] => 1 | 0]

▸ fsdistbind : forall A B : choiceType,

{dist A} -> (A -> {dist B}) -> {dist B}
def
= λb.∑a∈supp(d) d(a) × (f (a))(b) over ⋃x∈f (supp(d)) supp(x)

▸ fsdist_conv : forall A : choiceType,

prob -> {dist A} -> {dist A} -> {dist A}

def
= λa.p d1(a) + (1 − p)d2(a) over supp(d1) ∪ supp(d2)



The Start of a Hierarchy of Effects

probMonad

convex spaces

real cones

quasi-real cones

monad

functor

Solid arrow: inherits
Dotted arrow: uses



Probabilistic Program Verification using Rewriting

A biased coin with probability p:
Definition bcoin {M : probMonad} p : M bool := Ret T <| p |> Ret F.

Simple statement:
Definition two_coins p q : M (bool * bool) :=

do a <- bcoin p; do b <- bcoin q; Ret (a, b).

Lemma two_coinsE p q : two_coins p q = two_coins q p.

Proof:
rewrite /two_coins /bcoin.

(Ret T <|p|> Ret F) >>=

(fun a => (Ret T <|q|> Ret F) >>= (fun b => Ret (a, b)))

rewrite ![in LHS](choice_bindDl,bindretf).

(* choice_bindDl -> probability monad law *)

(* bindretf = ret x >= f = f x -> monad law *)

(Ret (T, T) <|q|> Ret (T, F)) <|p|> (Ret (F, T) <|q|> Ret (F, F))

rewrite -choiceACA.

(* interchange <|p|> <|q|> -> real cones *)

(Ret (T, T) <|p|> Ret (T, F)) <|q|> (Ret (F, T) <|p|> Ret (F, F))

...



Examples Formalized with The Monae Library

▸ tree relabeling [GH11], Spark aggregation [Mu19b], Monty-Hall
problem [GH11, Gib12]

▸ n-queens [GH11], completed by [Mu19a]
(we fixed an earlier version of the latter)

▸ quicksort [MC20]
(we completed a pre-existing formalization in Agda)

▸ Jaskelioff’s theory of modular monad transformers [Jas09]
(we actually proposed a fix for this theory)

Experiments documented in the following papers
[ANS19, AN21, AGNS21, SA22]



Combination of Monad Interfaces Can be Difficult

It was observed in [ASCG16] that [GH11] contains a mistake3:

▸ right-distributivity of bind over probabilistic choice
m≫= λx .(k x ◁ p▷ k ′ x) = (m≫= k)◁ p▷ (m≫= k ′)
combined with

▸ distributivity of probabilistic choice over arbitrary choice
m◁ p▷ (a2b) = (m◁ p▷ a)2 (m◁ p▷ b)

result in a degenerated theory:

▸ distributivity of arbitrary choice over probabilistic choice
m2 (a◁ p▷ b) = (m2 a)◁ p▷ (m2b)

▸ which implies a◁ p▷ b = a◁ q▷ b for all p,q ∈]0; 1[
[KP17, Thm A.3]

⇒ It is important to provide implementations for interfaces

3We checked with Monae that [GH11] was not relying on this mistake.



Hierarchy of Effects (cont’d)

Geometrically
convex

monad [Che17]

altCIMonad

probDrMonad

probMonad

convex spaces

real cones

quasi-real cones

altMonad

monad

functor

The probDrMonad adds:

▸ m≫= λx .(k x ◁ p▷ k ′ x) = (m≫= k)◁ p▷ (m≫= k ′)
The geometrically convex monad adds:

▸ m◁ p▷ (a2b) = (m◁ p▷ a)2 (m◁ p▷ b)



Model of the Geometrically Convex Monad
What is a computation in this monad?
▸ Gibbons observes that it should be a convex-closed sets of

probability distributions [Gib12]
▸ Cheung provides a construction using adjunctions between

categories [Che17]

We formalized Cheung’s construction [AGNS21]:

Type

CT
choiceType

CC
⊥

FC

UC

convType

CV
⊥

F0

U0

semiCompSemi-

-LattConvType

CS
⊥

F1

U1

This relies on an extension of Monae with concrete categories (to
go beyond Set)

Ask Takafumi here in this room! →

Statistical Model as Probabilistic Programs

▸ Example: guessing whether or not today’s a weekday by
looking at the number of buses passing by [Sta20]

normalize (

let x = sample (bernoulli (2 / 7)) in

let r = if x then 3 else 10 in

let _ = score (r ^ 4 / 4! * e ^ (- r)) in

return x)

▸ Intuitive explanation:
▸ sample takes a probability measure
▸ normalize returns a probability measure
▸ score (f x) means that we observe x from the distribution

corresponding to the density f
▸ here, observe 4 from the Poisson distribution

(of density rk

k!
e−r )

▸ Problem: existing formalizations in Coq use axioms
[HcS19, ZA22]



Formalization of Kernels using MathComp-Analysis

Staton proposed a semantics for programs with sampling, scoring,
and normalization using s-finite kernels [Sta17]

Definition:
▸ A kernel X ↝ Y is a function k ∶ X → ΣY → [0,∞] such that

1. for all x , k x is a measure
2. for all measurable set U, x ↦ k x U is measurable

Reminder: measure theory in MathComp-Analysis [AC22]
measurable spaces type measurableType

measure type {measure set T -> \bar R}

measurable functions predicate measurable_fun

Formal definition of kernel (notation R.-ker X ~> Y):

HB.mixin Record isKernel

X Y R (k : X -> {measure set Y -> \bar R}) :=

{ measurable_kernel :

forall U, measurable U -> measurable_fun setT (fun x => k x U) }.



S-Finite and Finite Kernels
A circular-looking definition

Definition:

▸ A kernel k ∶ X ↝ Y is finite when ∃r s.t. ∀x , k x Y < r
(uniformly upper bounded)

▸ A kernel k is s-finite when there exists a sequence of finite
kernels s such that k = ∑∞i=0 si

Circularity?

▸ s-finite kernels are more general than finite kernels
(so they should be defined first)

▸ finite kernels are needed to define s-finite kernels. . .



Wanted: Hierarchy of Kernels
To implement Staton’s semantics of probabilistic programs

Probability kernel X
prob

Y

Subprobability kernel X
subprob

Y

Finite kernel X
fin

Y

S-finite kernel X
s-fin

Y

Kernel X ↝ Y



S-Finite and Finite Kernels
A recipe using Hierarchy-Builder

1. Interface for s-finite kernels using a predicate for finite kernels:

HB.mixin Record Kernel_isSFinite_subdef

X Y R (k : X -> {measure set Y -> \bar R}) := {

sfinite_subdef : exists2 s : (R.-ker X ~> Y)^nat,

forall n, measure_fam_uub (s n) &

forall x U, measurable U -> k x U = kseries s x U }.

Notation: R.-sfker X ~> Y, inherits from R.-ker X ~> Y

2. Interface for finite kernels:

HB.mixin Record SFiniteKernel_isFinite

X Y R (k : X -> {measure set Y -> \bar R}) :=

{ measure_uub : measure_fam_uub k }.

Notation: R.-fker X ~> Y, inherits from R.-sfker X ~> X

3. Definitive interface for s-finite kernels:

HB.factory Record Kernel_isSFinite

X Y R (k : X -> {measure set Y -> \bar R})

of isKernel _ _ _ _ _ k := {

sfinite : exists s : (R.-fker X ~> Y)^nat,

forall x U, measurable U -> k x U = kseries s x U }.



Composition of S-finite Kernels

The main property of s-finite kernels is that they are stable by
composition (this provides a semantics for let x := e in e')

▸ Given l ∶ X ↝ Y and k ∶ X ×Y ↝ Z ,
the composition l ; k is defined by

λx U.∫
y
k (x , y)U(d l x)

▸ Reminder: integral theory in MathComp-Analysis [AC22]

∫x∈A f (x)(dµ) \int[mu]_(x in A) f x

▸ Formal definition of composition:

Definition kcomp l k x U := \int[l x]_y k (x, y) U.

▸ Staton proved that the composition of s-finite kernels is a s-finite
kernel [Sta17]. He skipped the proof that it is a kernel. It is not
trivial but it can be achieved it by adapting existing lemmas from
Fubini’s theorem available in MathComp-Analysis.



Semantics of Sampling using S-finite Kernels
For illustration

What is the semantics of sample (bernoulli (2 / 7))?

1. Build the measurable space of probability measures
pprobability Y R

▸ generated from the set of probability measures µ such that
µ(U) < r for all measurable sets U and 0 ≤ r ≤ 1

▸ The type X -> pprobability Y R is essentially
X -> {measure set Y -> \bar R}

2. P : X -> pprobability Y R is a kernel

▸ for any measurable set U, fun x => P x U is measurable

3. P : X -> pprobability Y R is a probability kernel

▸ because for all x, P x setT = 1
▸ it is therefore automatically a s-finite kernel

4. For our example, take for P the (constant) Bernoulli probability
measure (built out of Dirac measures)



Staton’s Buses in Coq

normalize (

let x = sample (bernoulli (2 / 7)) in

let r = if x then 3 else 10 in

let _ = score (r ^ 4 / 4! * e ^ (- r)) in

return x)

⇓

Definition kstaton_bus : R.-sfker T ~> mbool :=

letin (sample (bernoulli p27))

(letin

(letin (ite var2of2 (ret k3) (ret k10))

(score (measurable_fun_comp mh var3of3)))

(ret var2of3)).

(* NB: density function parameterized,

"De Bruijn indices" for variables *)

Definition staton_bus := normalize kstaton_bus.



Symbolic Evaluation of Statistical Models

We can evaluate a model to a distribution:

Lemma staton_busE P (t : R) U :

let N := ((2 / 7) * poisson4 3 + (5 / 7) * poisson4 10)%R in

staton_bus mpoisson4 P t U =

((2 / 7)%:E * (poisson4 3)%:E * \d_true U +

(5 / 7)%:E * (poisson4 10)%:E * \d_false U) * N^-1%:E.

(Proof by rewriting)
In mathematical notation:

2
7
34

4! e
−3

N
δ1 +

5
7
104

4! e
−10

N
δ0 = 0.780369δ1 + 0.219631δ0

So it is more likely that we are in the weekend



Commutativity Property of Probabilistic Programs
The main motivation for Staton’s work

Is the following program transformation correct?
let x := t in

let y := u in

ret (x, y)

↔ let y := u in

let x := t in

ret (x, y)

This is a consequence of Tonelli-Fubini’s theorem for s-finite
measures:

(* f measurable non-negative, m1, m2 s-finite *)

Lemma sfinite_fubini :

\int[m1]_x \int[m2]_y f (x, y) =

\int[m2]_y \int[m1]_x f (x, y).

(This is a consequence of Tonelli-Fubini’s theorem for σ-finite

measures—the one you find in a standard undergraduate textbook on

integration).



Conclusion
The MathComp project has been providing us with

▸ good tactic support
(e.g., rewrite in monadic equational reasoning)

▸ a rich, stable, flexible framework (we could combine MathComp
and the Coq standard library)

▸ libraries (finmap, MathComp-Analysis)

▸ methodologies (packed classes, naming conventions)

▸ tools (Hierarchy-Builder)

which let us

▸ develop original formalizations (InfoTheo, Monae)

▸ develop libraries for (probabilistic) program verification

▸ fix existing pencil-and-paper proofs

▸ retrofit results to MathComp (in particular
MathComp-Analysis)

